


Übungseinheit: # 5

KLICKOMANIA



Entwicklungsumgebung: Excel 2007

Kursunterlagen erstellt von: **TECHNISCHES BÜRO**

Ing. Harald Mitsch
Josefgasse 6
A 3380 Pöchlarn

MITSCH

Ihr IT-Betreuer in Pöchlarn
seit 1990 - 0676/588 09 16
<http://www.tb-mitsch.at>

letzte Aktualisierung: 11. Jänner 2019

INHALTSVERZEICHNIS

1.) Allgemeines	3
2.) Spielregeln	3
3.) Überlegungen zum Spielablauf	3
4.) Programmieren des Spiels	4
4.1.) Vorbereitungen	4
4.2.) Gestaltung des Spielfeldes	4
4.3.) Überlegungen zum Programmablauf	5
4.4.) Programmierung des Spieles	7
4.4.1.) Globale Konstanten und Variablen	7
4.4.2.) Funktion zum Starten des Spieles	8
4.4.3.) Funktion zur Steuerung nach Spielstart	10
4.4.4.) Brett nach links oder rechts bewegen	12
4.4.5.) Den Ball im Spielfeld bewegen	15
4.4.6.) Das Spiel mit Sounds hinterlegen	22
5.) Schlußbemerkungen	28

1.) Allgemeines

In dieser Übungseinheit soll das dynamische Verhalten von Zellen in Excel untersucht werden und wie wir das programmtechnisch lösen können. Das bedeutet, nach jedem Spielzug hat jede Zelle eine neue Bedeutung, eine neue Funktion oder ein anderes Verhalten, abhängig davon wie sich die Zelle vom vorherigen Spielzug verändert hat. Gleichzeitig wollen wir rekursive Funktionen erörtern. Dazu habe ich mir das Spiel **Klickomania** ausgesucht.

Die Datei **Click.exe** (ist zusätzlich im Ordner für die Übungseinheit #5 hinterlegt) ist bereits ein fertiges Spiel mit dem Sie sich mit der Funktionsweise vertraut machen können. Ein Doppelklick auf diese Datei genügt – es ist keine Installation notwendig.

2.) Spielregeln

Das vordefinierte Spielfeld, mit beliebiger Breite und Höhe, besteht aus quadratischen Zellen. Diese werden mit 4 bis 10 unterschiedlichen Farben zufällig eingefärbt. Klickt man auf eine Zelle, welche mindestens eine Nachbarzelle mit derselben Farbe hat, dann werden diese Zellen entfernt. Die darüberliegenden Zellen rutschen dann soweit nach unten, bis keine leeren Zellen mehr vorhanden sind – die leeren Zellen erscheinen jetzt am oberen Spielfeldrand. Grenzt an eine Zelle keine andere mit derselben Farbe an, dann bleibt der Klick wirkungslos. Nachdem eine Menge von Zellen entfernt wurde kommt es schließlich vor, daß eine Spalte (von oben bis unten) vollständig leer ist. In diesem Fall rücken die Zellen der Spalten rechts davon um eine Spalte nach links weiter. Die verbleibenden Zellen werden sozusagen in die linke untere Ecke zusammengeschoben. Ziel des Spieles ist es das gesamte Spielfeld zu leeren. Kann kein Klick mehr gemacht werden – es gibt keine zusammenhängende Zellen derselben Farbe mehr, dann hat der Spieler verloren.

3.) Überlegungen zum Spielablauf

Der Anwender soll nur auf eine Schaltfläche klicken können um das Spiel zu starten. Dabei soll nach jedem Start das Spielfeld neu aufgebaut und die Zellen mit zufälligen Farben befüllt werden. Zusätzlich soll auch ein zusammenhängender Rahmen über die Zellen mit der gleichen Farbe gezeichnet werden.

Beim Klick auf eine Zelle kontrollieren wir ob es Nachbarzellen mit derselben Farbe gibt. Wenn nicht, dann geben wir einen Sound aus, welcher einen Fehler signalisiert. Ansonst löschen wir alle zusammenhängenden Zellen. Dann lassen wir die verbliebenen Zellen nach unten rutschen und bei Bedarf schließen wir die Spalten nach links auf.

Bevor wir den nächsten Klick abfragen kontrollieren wir noch, ob das Spielfeld leer ist und ob es noch Zugmöglichkeiten gibt. Es das Spielfeld leer, dann spielen wir einen Sound für den Sieg ab, existiert kein Zug mehr, geben wir den Sound für eine Niederlage aus. Ansonst wird auf den nächsten Klick gewartet.

Ein Punktesystem soll einfach nur die gelöschten Zellen zählen und im Tabellenblatt ausgeben.

Hinweis: Bei Excel gibt es kein Ereignis welches beim Klicken auf eine Zelle aufgerufen wird. Daher verwenden wir das Ereignis: **SelectionChange**. Dazu müssen wir nach jedem Klick den Cursor (die Selektion) auf eine Zelle außerhalb des Spielfeldes setzen. Klickt man dann in das Spielfeld, so wird unsere Prozedur aufgerufen. Diese Prozedur wird aber auch dann aufgerufen, wenn wir über das Programm eine Zelle verändern (Farbe setzen, Rahmen zeichnen, ...). Daher müssen wir sicherstellen, daß unsere Prozedur nur ein einziges Mal aufgerufen wird! Wie? – das erkläre ich später.


4.) Programmieren des Spiels

4.1.) Vorbereitungen

Zuerst legen wir uns einen neuen Ordner an in welchem wir alle für das Spiel benötigte Dateien ablegen. Dann kopieren wir die Datei: **Muster_Vorlage.xlsm** (die überarbeitete Excel Datei aus der Übungseinheit #4) in diesen Ordner. Von derselben Datei legen wir uns eine weitere Kopie an und benennen sie um z.B. auf: **Klickomania.xlsm**. In dieser Datei werden wir das Spiel programmieren. Sollten wir dabei auf Funktionen oder andere Elemente stoßen, welche wir bei anderen Spielen ebenfalls gebrauchen könnten, dann werden wir diese zusätzlich auch in der Mustervorlage hinterlegen.

4.2.) Gestaltung des Spielfeldes

Um die Zellen im Tabellenblatt einfacher anklicken zu können (sie besser mit der Maus zu treffen) soll das Spielfeld beim Spielen mit einem Zoomfaktor von 200% dargestellt werden. Daher definieren wir die Spaltenbreite für die Spalten A bis S mit 2. Damit die Zellen quadratisch erscheinen setzen wir die Zeilenhöhe auf 12,75. Für die Schrift (auch wenn wir sie nicht benötigen), legen wir am besten den Font Arial mit einer Schriftgröße von 10 fest. Rechts neben dem Spielfeld sollen die Punkte und entsprechende Meldungen ausgegeben werden.



Nun benennen wir noch die Zelle für die Ausgabe der Punkte mit **PUNKTE**. Um den Cursor nach jedem Klick wieder aus dem Spielfeld zu nehmen definieren wir auch eine Zelle. Hier empfiehlt sich die Zelle mit dem Spieltitel ("Klickomania" in T1). Diese Zelle benennen wir mit **CURSOR_PARKPOSITION**. Eine weitere Zelle soll für die Ausgabe diverser Meldungen verwendet werden. Dazu verwende ich die Zelle T9 und hinterlege dort den Namen **MELDUNG**.

4.3.) Programmierung des Spieles

4.3.1.) Globale Konstanten und Variablen

Zuerst ändern wir Modul **Globales** folgende Konstanten:

```
Global Const PASSWORT = ""           'Passwort fuer Blattschutz.
Global Const PROGRAMMNAME = "Bricks" 'Name des Programmes.
Global Const TABELLENBLATT = "Bricks" 'Tabellenblatt fuer Spielfeld.
```

Dann hinterlegen wir noch alles was wir später beim Programmieren so benötigen werden:

```
Global Const TABELLENBLATT_LEVEL = "Level" 'Tabellenblatt fuer Levels -
                                           'ergaenzt mit 1, 2, ...
Global Const MAX_OBEN = 1                 '1. Zeile Spielfeld.
Global Const MAX_UNTEN = 44                'letzte Zeile Spielfeld.
Global Const MAX_LINKS = 1                 '1. Spalte Spielfeld (A).
Global Const MAX_RECHTS = 27               'letzte Spalte Spielfeld (AA).
Global Const MAX_BAELE = 10                'Maximale Anzahl der Baelle.
Global Const MAX_LEVEL = 5                 'Maximale Anzahl der Levels.

Global Const FARBE_HINTERGRUND = 16316664 '16316664 = weiss
Global Const FARBE_BRETT = 12611584        '12611584 = blau
Global Const FARBE_BALL = 255               '255 = rot

Global Const STARTGESCHWINDIGKEIT = 50     'in Millisekunden (50 = Standard)
Global Const GESCHWINDIGKEITSAENDERUNG = 2 'Nach jedem Level 2 ms kuerzer.
Global iGeschwindigkeit As Integer          'Aktuelle Geschwindigkeit.

Global stRichtung As String                 'Aktuelle Bewegungsrichtung Ball.
Global stRichtungNeu As String               'Neue Bewegungsrichtung Ball.
Global stTastendruck As String               'Zuletzt gedruckte Taste.

Global iZeileBrett As Integer                'Aktuelle Position fuer Brett:
Global iSpalteBrett As Integer               'Aktuelle Position fuer Ball:
Global iZeileBall As Integer
Global iSpalteBall As Integer
Global bBallImSpiel As Boolean               'Ball befindet sich im Spiel.

Global lPunkte As Long                      'Aktuelle Anzahl der Punkte.
Global iBaelle As Integer                   'Anzahl noch verfuegbare Baelle.
Global iLevel As Integer                    'Akt. Level (1 bis MAX_LEVEL).
Global iLevelIst As Integer                 'Akt. Level (1 bis ...).
Global stTabelleLevel As String              'Tabellenblattname akt. Level.
Global iAktBricks As Integer                 'Anzahl Ziegel im akt. Level.
```

4.4.2.) Funktion zum Starten des Spieles

Erstellen wir nun die Funktion **SpielStarten** im Modul **Funktionen** die folgendes erledigt: Zuerst den Blattschutz aufheben. Beim allerersten Aufruf der Startfunktion sollen die Standardwerte festgelegt und im Tabellenblatt ausgegeben werden. Bei jedem weiteren Aufruf wird unterschieden, ob ein neuer Level angezeigt werden soll oder ob nur ein neuer Ball ins Spiel gebracht wurde. Nach dem Einlesen eines neuen Levels muß die Anzahl der Ziegelsteine ermittelt werden, damit wir später feststellen können ob der Level bereits zu Ende gespielt wurde. Beim ersten Durchlauf der Levels 1 bis MAX_LEVELS soll das Brett nicht 3 sondern 5 Zellen breit sein. Zum Schluß wieder den Blattschutz setzen.

Langer Rede, kurzer Sinn – hier die fertige Funktion:

```
Public Function SpielStarten(bErsterStart As Boolean) As Boolean
'*****
' *      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      *
'*****
' *-----*
' *
' *      FUNKTION:      SpielStarten      *
' *
' *      BESCHREIBUNG:   Ist der Uebergabe-Parameter bErsterStart True, dann *
' *                     wird ein neuer Level eingeblendet. Ist die aktuelle *
' *                     Punkteanzahl auch 0, dann werden die Startwerte      *
' *                     festgelegt und im Tabellenblatt angezeigt.          *
' *                     Ist der Uebergabe-Parameter False, dann wird nur    *
' *                     ein neuer Ball auf dem Brett ausgegeben, die         *
' *                     aktuellen Ziegelsteine bleiben erhalten. Bei einem   *
' *                     neuen Level wird auch die Anzahl der vorhandenen     *
' *                     Bricks ermittelt.                                   *
' *
' *      RETURN         Boolean ..... True = Funktion in Ordnung          *
' *                     False = Fehler in Funktion                        *
' *
' *      PARAMETER:     Boolean ..... True = Neuen Level aufbauen          *
' *                     False = Nur Ball neu einblenden                  *
' *-----*
Dim iZeile          As Integer          'Aktuelle Zeilennummer.
Dim iSpalte         As Integer          'Aktuelle Spaltennummer.

On Error GoTo Fehler                    'Bei Fehler zur Fehlermarke.
SpielStarten = True                     'Returnwert = alles OK setzen.
bFluchttaste = False                   'ESC wurde noch nicht gedrueckt.

Set oTabellenblatt = Sheets(TABELLENBLATT) 'Akt. Tabellenblatt festlegen.
oTabellenblatt.Select                  'Tabellenblatt auswaehlen.
Call BlattschutzAufheben               'Blattschutz aufheben.

If bErsterStart = True And lPunkte = 0 Then 'Beim ersten Start alle Variablen
    lPunkte = 0                          'ruecksetzen, Standardwerte
    iBaelle = MAX_BAELE                  'festlegen fuer maximale Anzahl
    iLevel = 1                           'der Baelle, Level fuer Vorlage
    iLevelIst = 1                         'und aktuell gespielter Level.
    iGeschwindigkeit = STARTGESCHWINDIGKEIT 'Startgeschwindigkeit festlegen.
    oTabellenblatt.Range("PUNKTE") = lPunkte 'Ausgabe der Startwerte im
    oTabellenblatt.Range("LEVEL") = iLevel  'Tabellenblatt.
    oTabellenblatt.Range("BAELLE") = iBaelle
End If

bBallImSpiel = False                    'Ball ist noch nicht im Spiel.
iZeileBrett = MAX_UNTEN                  'Startpositionen fuer Brett und
iSpalteBrett = Int((MAX_RECHTS - MAX_LINKS) / 2) 'Ball mittig festlegen,
iZeileBall = iZeileBrett - 1             'Ball mittig und ueber Brett).
iSpalteBall = iSpalteBrett + 1

If Int((2 * Rnd) + 1) = 1 Then           'Zufaellige Richtung festlegen,
    stRichtung = "LO"                   'ob der Ball beim Einwerfen
    stRichtungNeu = "LO"                'nach links oben
Else                                     'oder
    stRichtung = "RO"                   'nach rechts oben
```

```

stRichtungNeu = "RO"
End If

stTabelleLevel = TABELLENBLATT_LEVEL & Trim(Str(iLevel)) 'Tabellenblatt fuer die
'wegspringen soll. Natuerlich
'ist das auch die neue Richtung.

For iZeile = MAX_UNTEN - 1 To MAX_UNTEN
    For iSpalte = MAX_LINKS To MAX_RECHTS
        oTabellenblatt.Cells(iZeile, iSpalte).Interior.Color = _
            Sheets(stTabelleLevel).Cells(iZeile, iSpalte).Interior.Color
    Next iSpalte
Next iZeile

If bErsterStart = True Then
    iAktBricks = 0
    For iZeile = MAX_OBEN To MAX_UNTEN - 2
        For iSpalte = MAX_LINKS To MAX_RECHTS
            oTabellenblatt.Cells(iZeile, iSpalte).Interior.Color = _
                Sheets(stTabelleLevel).Cells(iZeile, iSpalte).Interior.Color
            If oTabellenblatt.Cells(iZeile, iSpalte).Interior.Color <> FARBE_HINTERGRUND Then
                iAktBricks = iAktBricks + 1
            End If
        Next iSpalte
    Next iZeile
    iAktBricks = Int(iAktBricks / 3)
End If

If iLevelIst <= 5 Then
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett - 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 3).Interior.Color = FARBE_BRETT
End If
'Ziegelsteine uebertragen:
'Anzahl der Ziegel im Level
'ruecksetzen. Dann das gesamte
'Spielfeld abarbeiten und
'dann Anzahl der gefaerbten
'Zellen ermitteln. Da jeder
'Ziegel aus 3 Zellen besteht
'errechnet sich die Anzahl
'der Ziegel mit Division / 3.

Range("CURSOR_PARKPOSITION").Select
Call BlattschutzSetzen
'Cursor auf Park-Position.
'Blattschutz setzen.

Beenden:
    Application.Cursor = xlDefault
    Exit Function
'Fehlerbehandlungsroutine:
'=====
'Sanduhr ausblenden.
'Funktion abbrechen.

Fehler:
    SpielStarten = False
    MsgBox Err.Description
    Resume Beenden
    Resume
End Function
'Fehlermarke - Fehlerbehandlung:
'Returnwert = Fehler setzen.
'Fehlermeldung ausgeben.
'Bei Funktionsende weitermachen.
'Nur fuer Testzwecke hinterlegt.
'Funktionsende.

```

4.4.3.) Funktion zur Steuerung nach Spielstart

Als nächstes erstellen wir nun die Funktion **Spielen** im Modul **Funktionen** welche die Ablaufsteuerung während des Spielens übernimmt. Dazu müssen wir uns vorab noch folgende leere Funktionen anlegen:

```

Public Function NachRechts()
End Function
'Funktionsende.

Public Function NachLinks()
End Function
'Funktionsende.

```

```
Public Function BallBewegen()  
End Function                                     'Funktionsende.
```

Die Spielsteuerung könnte dann so aussehen:

```
Public Function Spielen()  
'*****  
'*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      */  
'*****  
'*-----*/  
'*      */  
'*      FUNKTION:      Spielen      */  
'*      */  
'*      BESCHREIBUNG:   Startet eine Endlos-Schleife welche die Tastatur      */  
'*                      ueberwacht und bei entsprechendem Tastendruck die      */  
'*                      dazugehoerige Funktion aufruft. Mit der Taste PAUSE */  
'*                      kann die Endlos-Schleife vorzeitig beendet werden.      */  
'*                      Der stTastendruck = "SPIELEND" wird nachdem das      */  
'*                      Spiel zu Ende ist in der entsprechenden Funktion      */  
'*                      gesetzt: BallBewegen      */  
'*      */  
'*      RETURN          nichts      */  
'*      */  
'*      PARAMETER:      keine      */  
'*-----*/  
On Error GoTo Fehler                                     'Bei Fehler zur Fehlerbehandlung.  
  
stTastendruck = ""                                     'Gedruckte Taste ruecksetzen.  
While stTastendruck = ""                               'Solange bis Taste gedrueckt wird  
    Range("CURSOR_PARKPOSITION").Select                'Cursor zurueck auf Park-Position.  
  
    If GetAsyncKeyState(vbKeyEscape) Then              'Wird ESC-Taste gedrueckt, dann  
        stTastendruck = "ESC"                          'Abbruchsbedingung fuer Endlos-  
        On Error Resume Next                          'Schleife festlegen und bei  
    End If                                             'Fehler einfach weitermachen.  
  
    If GetAsyncKeyState(vbKeyPause) Then                'WennPAUSE-Taste gedrueckt, dann  
        stTastendruck = "ABBRECHEN"                  'Abbruchsbedingung fuer Endlos-  
    End If                                             'Schleife festlegen.  
  
    If GetAsyncKeyState(vbKeyRight) Then                'Wird RECHTS-Taste gedrueckt:  
        Call NachRechts                               'Brett nach rechts bewegen.  
    End If  
  
    If GetAsyncKeyState(vbKeyLeft) Then                 'Wird LINKS-Taste gedrueckt, dann  
        Call NachLinks                                'Brett nach links bewegen.  
    End If  
  
    If GetAsyncKeyState(vbKeyUp) Then                   'Wird UP-Taste gedrueckt, dann  
        bBallImSpiel = True                           'Ball ins Spiel einwerfen.  
    End If  
  
    If bBallImSpiel = True Then                         'Wenn Ball im Spiel ist, dann  
        Call BallBewegen                             'den Ball bewegen.  
    End If  
    DoEvents                                           'Dem Programm zusaetzliche  
                                                    'Rechenzeit zur Verfuegung
```



```
Sleep iGeschwindigkeit  
Wend
```

```
If stTastendruck = "ESC" Then  
    Call Fluchttaste  
End If
```

```
If stTastendruck = "ABBRECHEN" Then  
    MsgBox "Spiel wurde abgebrochen!", _  
        vbOKOnly + vbCritical + vbDefaultButton1, PROGRAMMNAME  
End If
```

```
If stTastendruck = "SPIELEND" Then  
    MsgBox "SPIELEND - keine Bälle mehr!", _  
        vbOKOnly + vbCritical + vbDefaultButton1, PROGRAMMNAME  
End If
```

```
Beenden:  
    Application.Cursor = xlDefault  
    Exit Function
```

```
Fehler:  
    MsgBox Err.Description  
    Resume Beenden  
Resume  
End Function
```

```
'stellen, da sonst alles  
'nach ca. 4 Sekunden abstuerzt!  
'Eine Zeitlang warten (50 ms).  
'(Steuert Spielgeschwindigkeit.)
```

```
'Bei ESC-Taste.  
'Spiel sofort beenden und Datei  
'ohne Speichern schliessen.
```

```
'Bei PAUSE-.Taste:
```

```
'Bei Spielende:
```

```
'Fehlerbehandlungsroutine:  
'=====
```

```
'Sanduhr ausblenden.  
'Funktion abbrechen.
```

```
'Fehlermarke - Fehlerbehandlung:  
'Fehlermeldung ausgeben.  
'Bei Funktionsende weitermachen.  
'Nur fuer Testzwecke hinterlegt.  
'Funktionsende.
```

Das Grundgerüst für das Spiel haben wir jetzt ausprogrammiert. Jetzt brauchen wir nur mehr dafür sorgen, daß die beiden Funktionen auch wunschgemäß aufgerufen werden. Zuerst in der Prozedur **Workbook_Open** im Modul **DieseArbeitsmappe**. In dieser erweitern wir den bisherigen Funktionsaufruf folgendermaßen:

```
'*-----*/
1Punkte = 0                                'Punkteanzahl 0 setzen.
Call SpielStarten(True)                    'Startprogramm aufrufen.
'*-----*/
```

Dann benötigen wir noch ein Makro (im Modul **Makros**) welche ebenfalls das Spiel startet:

```
Public Sub Spiel_Starten()
'*****/
'* Makro wird ausgefuehrt bei Klick auf Schaltflaeche: Starten */
'*****/
1Punkte = 0                                'Punkte auf 0 setzten.
Call SpielStarten(True)                    'Spielfeld aufbauen.
Call Spielen                               'Spiel starten.
End Sub                                     'Prozedurende.
```

Das Makro **Spiel_Starten** weisen wir abschließend der Schaltfläche **Starten** zu.

Nun können wir die Datei speichern, schließen und erneut öffnen – und jetzt testen... Beim Testen bzw. während wir das Spiel weiter ausprogrammieren empfiehlt es sich die Ansicht auf Zoom 75% zu schalten und die Multifunktionsleiste zu minimieren bzw. manuellen Zoomfaktor auf 70% zu setzen.

4.4.4.) Brett nach rechts oder links bewegen

Als nächstes erweitern wir im Modul **Funktionen** die Funktionen **NachRechts** und **NachLinks**: Die Dokumentation im Programmcode sollte die Funktionsweise klar beschreiben.

```
Public Function NachRechts ()
'*****/
'* INTERNES UNTERPROGRAMM - INTERNE FUNKTION */
'*****/
'*-----*/
'* */
'* FUNKTION: NachRechts */
'* */
'* BESCHREIBUNG: Bewegt das Brett um eine Position nach rechts. Wenn */
'* der Ball nicht im Spiel ist, er liegt auf dem */
'* Brett auf, dann den Ball auch eine Position nach */
'* rechts verschieben. */
'* Gleichzeitig wird darauf geachtet, dass das Brett */
'* nicht über das Spielfeld hinaus verschoben werden */
'* kann. */
'* */
'* RETURN nichts */
'* */
'* PARAMETER: keine */
'* */
'*-----*/
On Error GoTo Fehler                    'Bei Fehler zur Fehlerbehandlung.
```

```

If iLevelIst <= 5 Then                                'Brett ist 5 Zellen breit:
    'Aktuelle 5 Zellen mit Hintergrundfarbe loeschen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett - 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 3).Interior.Color = FARBE_HINTERGRUND

    'Spaltenbegrenzung um nicht ueber den rechten Spielfeldrand zu kommen:
    If iSpalteBrett < MAX_RECHTS - 3 Then iSpalteBrett = iSpalteBrett + 1

    'Neue 5 Zellen fuer Brett mit Brettfarbe zeichnen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett - 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 3).Interior.Color = FARBE_BRETT
Else                                                    'Brett ist 3 Zellen breit:
    'Die 3 aktuellen Zellen fuer Brett mit Hintergrundfarbe loeschen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_HINTERGRUND

    'Spaltenbegrenzung um nicht ueber den rechten Spielfeldrand zu kommen:
    If iSpalteBrett < MAX_RECHTS - 2 Then iSpalteBrett = iSpalteBrett + 1

    'Neue 3 Zellen fuer Brett mit Brettfarbe zeichnen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_BRETT
End If

If bBallImSpiel = False Then                            'Wenn der Ball nicht im Spiel ist,
    oTabellenblatt.Cells(iZeileBrett - 1, iSpalteBrett + 0).Interior.Color = FARBE_HINTERGRUND    'dann auch Ball eins nach rechts.
    oTabellenblatt.Cells(iZeileBrett - 1, iSpalteBrett + 1).Interior.Color = FARBE_BALL
    iSpalteBall = iSpalteBrett + 1
    iZeileBall = iZeileBrett - 1
    'Neue Zeile und Spalte fuer
    'Ballposition merken.
End If
Range("CURSOR_PARKPOSITION").Select                    'Cursor auf Parkposition setzen.

Beenden:
    Application.Cursor = xlDefault
    Exit Function

Fehler:
    NachRechts = False
    MsgBox Err.Description
    Resume Beenden
    Resume
End Function

Public Function NachLinks()
    '*****
    '*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      *
    '*****
    '*-----*
    '*
    '*      FUNKTION:      NachLinks
    '*
    '*      BESCHREIBUNG:  Bewegt das Brett um eine Position nach links. Wenn
    '*                      der Ball nicht im Spiel ist, er liegt auf dem
    '*                      Brett auf, dann den Ball auch eine Position nach
    '*                      links verschieben.
    '*                      Gleichzeitig wird darauf geachtet, dass das Brett
    '*
    '*****

```

```

'*          nicht über das Spielfeld hinaus verschoben werden */
'*          kann.                                           */
'*                                                     */
'*          RETURN          nichts                        */
'*                                                     */
'*          PARAMETER:      keine                        */
'*                                                     */
'*-----*/
On Error GoTo Fehler                                     'Bei Fehler zur Fehlerbehandlung.

If iLevelIst <= 5 Then                                   'Brett ist 5 Zellen breit:
    'Aktuelle 5 Zellen mit Hintergrundfarbe loeschen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 3).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett - 1).Interior.Color = FARBE_HINTERGRUND

    'Spaltenbegrenzung um nicht ueber den linken Spielfeldrand zu kommen:
    If iSpalteBrett > MAX_LINKS + 1 Then iSpalteBrett = iSpalteBrett - 1

    'Neue 5 Zellen fuer Brett mit Brettfarbe zeichnen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett - 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 3).Interior.Color = FARBE_BRETT
Else
    'Brett ist 3 Zellen breit:
    'Die 3 aktuellen Zellen fuer Brett mit Hintergrundfarbe loeschen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_HINTERGRUND

    'Spaltenbegrenzung um nicht ueber den linken Spielfeldrand zu kommen:
    If iSpalteBrett > MAX_LINKS Then iSpalteBrett = iSpalteBrett - 1

    'Neue 3 Zellen fuer Brett mit Brettfarbe zeichnen:
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 0).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 1).Interior.Color = FARBE_BRETT
    oTabellenblatt.Cells(iZeileBrett, iSpalteBrett + 2).Interior.Color = FARBE_BRETT
End If

If bBallImSpiel = False Then
    'Wenn der Ball nicht im Spiel ist,
    'dann auch Ball eins nach links.
    oTabellenblatt.Cells(iZeileBrett - 1, iSpalteBrett + 2).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeileBrett - 1, iSpalteBrett + 1).Interior.Color = FARBE_BALL
    iSpalteBall = iSpalteBrett + 1
    'Neue Zeile und Spalte fuer
    iZeileBall = iZeileBrett - 1
    'Ballposition merken.
End If
Range("CURSOR_PARKPOSITION").Select                    'Cursor auf Parkposition setzen.

Beenden:
    Application.Cursor = xlDefault
    Exit Function
    'Fehlerbehandlungsroutine:
    '=====
    'Sanduhr ausblenden.
    'Funktion abbrechen.

Fehler:
    'Fehlermarke - Fehlerbehandlung:
    NachLinks = False
    'Returnwert = Fehler setzen.
    MsgBox Err.Description
    'Fehlermeldung ausgeben.
    Resume Beenden
    'Bei Funktionsende weitermachen.
    Resume
    'Nur fuer Testzwecke hinterlegt.
End Function
    'Funktionsende.

```

Speichern, schließen und öffnen Sie die Datei erneut. Testen Sie nun ob sich das Brett mit den vorgesehenen Tasten in der beabsichtigten Art und Weise bewegen läßt.
Sieht ja schon ganz aus... - oder ?

4.4.5.) Den Ball im Spielfeld bewegen

Dazu erweitern wir im Modul **Funktionen** die Funktion **BallBewegen**: Legen wir uns vorerst einmal nur die Struktur der Funktion an. Zuerst löschen wir die aktuelle Ballposition indem wir diese Zelle mit der Hintergrundfarbe füllen. Dann entscheiden wir anhand der aktuellen Bewegungsrichtung, auf welche neue Position der Ball zu bewegen ist. Dann kontrollieren wir, ob sich die neue Position noch im Spielfeld befindet – besser gesagt, wir schauen uns an, ob sich die neue Position am Spielfeldrand befindet. Wenn ja, dann machen wir einen Richtungswechsel je nachdem welcher Rand getroffen wurde.

```
Public Function BallBewegen()  
'*****  
'*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      */  
'*****  
'*-----*/  
'*      */  
'*      FUNKTION:      BallBewegen      */  
'*      */  
'*      BESCHREIBUNG:  Bewegt den Ball um eine Position weiter. Aendert am */  
'*      Rand die Richtung. Trifft Ball auf Ziegel dann wird */  
'*      die Funktion zum Loeschen des Ziegels aufgerufen.  */  
'*      */  
'*      RETURN      nichts      */  
'*      */  
'*      PARAMETER:    keine      */  
'*      */  
'*-----*/  
On Error GoTo Fehler      'Bei Fehler zur Fehlerbehandlung.  
                          'Aktuellen Ball loeschen.  
oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_HINTERGRUND  
  
Select Case stRichtung      'Je nach Bewegungsrichtung:  
Case "RO"                  'Ball bewegt sich nach: rechts oben  
    iZeileBall = iZeileBall - 1      'Neue Ballposition und  
    iSpalteBall = iSpalteBall + 1    'neue Bewegungsrichtung festlegen.  
    If iZeileBall <= MAX_OBEN Then iZeileBall = MAX_OBEN: stRichtungNeu = "RU"  
    If iSpalteBall >= MAX_RECHTS Then iSpalteBall = MAX_RECHTS: stRichtungNeu = "LO"  
  
Case "LO"                  'Ball bewegt sich nach: links oben  
    iZeileBall = iZeileBall - 1      'Neue Ballposition und  
    iSpalteBall = iSpalteBall - 1    'neue Bewegungsrichtung festlegen.  
    If iZeileBall <= MAX_OBEN Then iZeileBall = MAX_OBEN: stRichtungNeu = "LU"  
    If iSpalteBall <= MAX_LINKS Then iSpalteBall = MAX_LINKS: stRichtungNeu = "RO"  
  
Case "RU"                  'Ball bewegt sich nach: rechts unten  
    iZeileBall = iZeileBall + 1      'Neue Ballposition und  
    iSpalteBall = iSpalteBall + 1    'neue Bewegungsrichtung festlegen.  
    If iZeileBall >= MAX_UNTEN Then iZeileBall = MAX_UNTEN: stRichtungNeu = "RO"  
    If iSpalteBall >= MAX_RECHTS Then iSpalteBall = MAX_RECHTS: stRichtungNeu = "LU"  
  
Case "LU"                  'Ball bewegt sich nach: links unten  
    iZeileBall = iZeileBall + 1      'Neue Ballposition und  
    iSpalteBall = iSpalteBall - 1    'neue Bewegungsrichtung festlegen.  
    If iZeileBall >= MAX_UNTEN Then iZeileBall = MAX_UNTEN: stRichtungNeu = "LO"  
    If iSpalteBall <= MAX_LINKS Then iSpalteBall = MAX_LINKS: stRichtungNeu = "RU"  
End Select  
  
oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BALL  
stRichtung = stRichtungNeu      'Ball auf neue Position setzen:  
Range("CURSOR_PARKPOSITION").Select      'Neue Richtung festlegen.  
                                          'Cursor auf Parkposition setzen.  
  
Beenden:                      'Fehlerbehandlungsroutine:  
    Application.Cursor = xlDefault      '=====  
    Exit Function                      'Sanduhr ausblenden.  
                                          'Funktion abbrechen.
```

```
Fehler:
NachLinks = False
MsgBox Err.Description
Resume Beenden
Resume
End Function

'Fehlermarke - Fehlerbehandlung:
'Returnwert = Fehler setzen.
'Fehlermeldung ausgeben.
'Bei Funktionsende weitermachen.
'Nur fuer Testzwecke hinterlegt.
'Funktionsende.
```

Speichern Sie die Datei und testen die Bewegung des Balles. Sieht ja schon ganz gut aus (auch wenn der Ball noch über die Ziegel bewegt wird). Jetzt müssen wir natürlich am unteren Spielfeldrand noch abfragen, ob sich unter dem Ball auch das Brett befindet:

Diese Kontrolle benötigen wir nur dann, wenn sich der Ball nach unten bewegt, also bei den Richtungen **RU** und **LU**. Dabei gehen wir folgendermaßen vor: Zuerst berechnen wir, wie bereits erledigt, die neue Position des Balles. Befindet sich die neue Position in der Zeile des Brettes, dann kontrollieren wir, ob die neue Position die Hintergrundfarbe des Brettes hat. Wenn ja, dann machen wir die letzte Positionsänderung wieder rückgängig. Den Richtungswechsel selbst haben wir bereits im bisherigen Programmcode festgelegt.

Befindet sich jedoch kein Brett unter dem Ball, so ist der Ball aus dem Spiel. Die Anzahl der verfügbaren Bälle wird um eins verringert. Um die aktuelle Anzahl der Bälle im Tabellenblatt zu aktualisieren müssen wir zuerst den Blattschutz aufheben. Dann aktualisieren wir das Tabellenblatt und setzen abschließend den Blattschutz erneut.

Gleichzeitig können wir hier auch überprüfen ob das Spielende erreicht wurde – nämlich dann, wenn keine Bälle mehr im Spiel sind. In diesem Fall setzen wir eine Abbruchsbedingung (in der Funktion **Spielen** muß die Variable **Tastendruck** ungleich "" sein). Anhand dieser Variable können wir in der Funktion **Spielen** auch eine genauere Abschlußmeldung ausgeben (ABBRECHEN bzw. SPIELEND).

Sind noch Bälle im Spiel, dann rufen wir die Funktion **SpielStarten** mit dem Parameter **False** auf. Dabei wird ein neuer Ball ins Spiel gebracht. Der Ball wird dabei auf das Brett gelegt und das Brett samt Ball wird in der Mitte des Spielfeldes ausgegeben. Die aktuelle Anordnung der Ziegel bleibt erhalten – man kann also mit dem Spiel fortfahren ohne ganz neu von vorne beginnen zu müssen.

Nun die Umsetzung im Programmcode:

```
Case "RU"
iZeileBall = iZeileBall + 1
iSpalteBall = iSpalteBall + 1
If iZeileBall >= MAX_UNTEN Then iZeileBall = MAX_UNTEN: stRichtungNeu = "RO"
If iSpalteBall >= MAX_RECHTS Then iSpalteBall = MAX_RECHTS: stRichtungNeu = "LU"

'Ball trifft auf Brett oder nicht:
If iZeileBall = MAX_UNTEN Then
If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BRETT Then
iZeileBall = iZeileBall - 1
iSpalteBall = iSpalteBall - 1
Else
bBallImSpiel = False
iBaele = iBaele - 1
Call BlattschutzAufheben
oTabellenblatt.Range("BAELLE") = iBaele

Call BlattschutzSetzen
If iBaele > 0 Then
Call SpielStarten(False)
Else
stTastendruck = "SPIELEND"
End If
End If

'Ball bewegt sich nach: rechts unten
'Neue Ballposition und
'neue Bewegungsrichtung festlegen.
'Ball in Zeile von Brett:
'Brett ist unter Ball:
'Letzte Ballbewegung
'rueckgaengig machen.
'Sonst ist Ball aus dem Spielfeld.
'Anzahl der Baele verringern.
'Blattschutz aufheben und
'im Tabellenblatt ausgeben.
'Blattschutz wieder setzen.
'Blattschutz setzen.
'Sind noch Baele im Spiel, dann
'neuen Ball auf Brett legen.
'Wenn kein Ball mehr im Spiel:
'Abbruchsbedingung fuer
'Endlos-Schleife setzen.
```


End If

```
Case "LU"
    iZeileBall = iZeileBall + 1
    iSpalteBall = iSpalteBall - 1
    If iZeileBall >= MAX_UNTEN Then iZeileBall = MAX_UNTEN: stRichtungNeu = "LO"
    If iSpalteBall <= MAX_LINKS Then iSpalteBall = MAX_LINKS: stRichtungNeu = "RU"

    'Ball trifft auf Brett oder nicht:
    If iZeileBall = MAX_UNTEN Then
        If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BRETT Then
            iZeileBall = iZeileBall - 1
            iSpalteBall = iSpalteBall + 1
        Else
            bBallImSpiel = False
            iBaelle = iBaelle - 1
            Call BlattschutzAufheben
            oTabellenblatt.Range("BAELLE") = iBaelle

            Call BlattschutzSetzen
            If iBaelle > 0 Then
                Call SpielStarten(False)
            Else
                stTastendruck = "SPIELEND"
            End If
        End If
    End If
End If
```

'Ball bewegt sich nach: links unten
'Neue Ballposition und
'neue Bewegungsrichtung festlegen.
'Ball in Zeile von Brett:
'Brett ist unter Ball:
'Letzte Ballbewegung
'rueckgaengig machen.
'Sonst ist Ball aus dem Spielfeld.
'Anzahl der Baelle verringern.
'Blattschutz aufheben und
'im Tabellenblatt ausgeben.
'Blattschutz wieder setzen.
'Blattschutz setzen.
'Sind noch Baelle im Spiel, dann
'neuen Ball auf Brett legen.
'Wenn kein Ball mehr im Spiel:
'Abbruchsbedingung fuer
'Endlos-Schleife setzen.

Jetzt fehlt uns nur noch ein Programmteil, nämlich das Entfernen eines Ziegels nachdem der Ball ihn getroffen hat. Dabei soll es egal sein, ob der Ball von unten, von oben oder von der Seite kommt (gemeint ist damit, die Eckpunkte von Ball und Ziegel treffen sich). Dazu werden wir anschließend noch die neue Funktion **ZiegelLoeschen** erstellen.

Grundsätzlich kann ein Ziegel aus allen Richtungen getroffen werden und nach jedem Treffer muß auch die Richtung des Balles geändert werden. Daher ist der anschließende Code bei allen Richtungen - mit entsprechender Adaptierung - zu erweitern. Hier der Programmcode nur für die Richtung **RO** (der vollständige Code für die Funktion BallBewegen siehe am Ende dieses Kapitels):

```
'Ziegel wurde getroffen:
If iZeileBall >= MAX_OBEN Then
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND Then
        Call ZiegelLoeschen(iZeileBall, iSpalteBall)
        stRichtungNeu = "RU"
    End If
End If
If iZeileBall > MAX_OBEN Then
    If oTabellenblatt.Cells(iZeileBall - 1, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND Then
        Call ZiegelLoeschen(iZeileBall - 1, iSpalteBall)
        stRichtungNeu = "RU"
    End If
End If
```

'Kontrolle ob die neue Position
'Teil eines Ziegels ist:
'Wenn ja, dann Ziegel entfernen
'und die Richtung aendern.
'Zusatzkontrolle falls Ziegel
'von der Seite getroffen wird:
'Wenn ja, Ziegel entfernen
'und die Richtung aendern.

Bevor wir das Programm testen können müssen wir noch die Funktion **ZiegelLoeschen** im Modul **Funktionen** erstellen. Diese Funktion ersetzt die Ziegelfarbe mit der Hintergrundfarbe anhand der neuen Position des Spielballes. Um die korrekten Zellen für einen Ziegel zu finden benutzen wir die Funktion **Mod** (Rest einer Division).

Gleichzeitig erhöhen wir in dieser Funktion die Punkteanzahl und kontrollieren, ob noch Ziegelsteine im aktuellen Level vorhanden sind. Wenn nicht, dann lassen wir den nächsten Level anzeigen und erhöhen die Geschwindigkeit des Balles, indem wir die Wartezeit für das Bewegen des Balles verkleinern.

Sicherheitshalber begrenzen wir die kürzeste Wartezeit auf 5 Millisekunden. Vor der Aktualisierung der Punkte und des Levels muß natürlich der Blattschutz zuvor deaktiviert und danach wieder aktiviert werden.

Abschließend rufen wir die Funktion **SpielStarten** mit dem Parameter **True** auf; dabei wird der neue Level ins Spielfeld einkopiert. Der Ball wird dabei auf das Brett gelegt und das Brett samt Ball wird in der Mitte des Spielfeldes ausgegeben.

Hier nun die fertige Funktion:

```
Public Function ZiegelLoeschen(iZeile As Integer, iSpalte As Integer)
'*****
' *      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      *
'*****
' *-----*
' *
' *      FUNKTION:      ZiegelLoeschen      *
' *
' *      BESCHREIBUNG:  Loescht den Ziegel auf den der Ball getroffen ist *
' *                      und erhoeht die Punkteanzahl.      *
' *
' *      RETURN      nichts      *
' *
' *      PARAMETER:    Integer ..... Aktuelle Zeile des Balles      *
' *                      Integer ..... Aktuelle Spalte des Balles      *
' *-----*
On Error GoTo Fehler                                'Bei Fehler zur Fehlerbehandlung.

If iSpalteBall Mod 3 = 0 Then                        'Ball trifft letzte Zelle:
    oTabellenblatt.Cells(iZeile, iSpalte - 2).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte - 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte + 0).Interior.Color = FARBE_HINTERGRUND
ElseIf iSpalteBall Mod 3 = 1 Then                    'Ball trifft erste Zelle:
    oTabellenblatt.Cells(iZeile, iSpalte + 0).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte + 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte + 2).Interior.Color = FARBE_HINTERGRUND
ElseIf iSpalteBall Mod 3 = 2 Then                    'Ball trifft mittlere Zelle:
    oTabellenblatt.Cells(iZeile, iSpalte - 1).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte + 0).Interior.Color = FARBE_HINTERGRUND
    oTabellenblatt.Cells(iZeile, iSpalte + 1).Interior.Color = FARBE_HINTERGRUND
End If                                                'Entsprechende Ziegelzellen
                                                    'mit Hintergrundfarbe loeschen.

Call BlattschutzAufheben                            'Blattschutz aufheben.
lPunkte = lPunkte + 10                               'Einfache Punktebewertung:
oTabellenblatt.Range("PUNKTE") = lPunkte             'Punkte ausgeben.
iAktBricks = iAktBricks - 1                           'Anzahl Ziegeln verringern.
If iAktBricks <= 0 Then                               'Keine Ziegeln mehr vorhanden:
    iLevelIst = iLevelIst + 1                         'Tatsaechlichen Level erhoehen.
    iLevel = iLevel + 1                               'Level erhoehen, falls max.
    If iLevel > MAX_LEVEL Then iLevel = 1              'Level ueberschritten, dann
                                                    'wieder mit Level 1 beginnen.
    oTabellenblatt.Range("LEVEL") = iLevelIst         'Geschwindigkeit erhoehen:
    iGeschwindigkeit = iGeschwindigkeit - GESCHWINDIGKEITSAENDERUNG 'jedoch nicht
    If iGeschwindigkeit < 5 Then iGeschwindigkeit = 5 'unter 5 Millisekunden.
    Call SpielStarten(True)                          'Neuen Level anzeigen.
Else                                                  'Blattschutz wieder setzen.
    Call BlattschutzSetzen                            'Blattschutz setzen.
```


End If

Beenden:

```
Application.Cursor = xlDefault
Exit Function
```

```
'Fehlerbehandlungsroutine:
'=====
'Sanduhr ausblenden.
'Funktion abbrechen.
```

Fehler:

```
ZiegelLoeschen = False
MsgBox Err.Description
Resume Beenden
Resume
```

```
'Fehlermarke - Fehlerbehandlung:
'Returnwert = Fehler setzen.
'Fehlermeldung ausgeben.
'Bei Funktionsende weitermachen.
'Nur fuer Testzwecke hinterlegt.
'Funktionsende.
```

End Function

Jetzt aber, wie zuvor schon angesprochen der vollständige Programmcode der Funktion **BallBewegen**: In diesem Programmcode ist bereits die Hinterlegung von Sounds, welche wir erst im nächsten Kapitel abhandeln, enthalten. Diese Codezeilen sind gelb hinterlegt und werden später näher erklärt. Um das Spiel vorab zu testen sind die gelben Programmzeilen unter Kommentar zu setzen.

Public Function BallBewegen()

```
'*****
'*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION
'*
'*-----
'*
'*      FUNKTION:      BallBewegen
'*
'*      BESCHREIBUNG:  Bewegt den Ball um eine Position weiter. Aendert am
'*                      Rand die Richtung. Trifft Ball auf Ziegel dann wird
'*                      die Funktion zum Loeschen des Ziegels aufgerufen.
'*
'*                      Trifft Ball nicht auf Brett, dann wird ein Ball
'*                      abgezogen. Sind keine Baele mehr vorhanden, dann
'*                      ist das Spielende erreicht.
'*
'*      RETURN         nichts
'*
'*      PARAMETER:     keine
'*-----
'*
```

```
Dim bSoundWand      As Boolean      'Hilfsvariable fuer Sound.
Dim stSoundAlias    As String      'Wechen Sound abspielen.
```

On Error GoTo Fehler 'Bei Fehler zur Fehlerbehandlung.

bSoundWand = False 'Sound Wand nicht ausgegeben.

stSoundAlias = "" 'Sound fuer Alias ruecksetzen.

'Aktuellen Ball loeschen.

oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_HINTERGRUND

Select Case stRichtung

Case "RO"

iZeileBall = iZeileBall - 1

iSpalteBall = iSpalteBall + 1

If iZeileBall <= MAX_OBEN Then

iZeileBall = MAX_OBEN

stRichtungNeu = "RU"

stSoundAlias = ALIAS_SOUND_WAND

End If

If iSpalteBall >= MAX_RECHTS Then

iSpalteBall = MAX_RECHTS

stRichtungNeu = "LO"

stSoundAlias = ALIAS_SOUND_WAND

End If

'Je nach Bewegungsrichtung:

'Ball bewegt sich nach: rechts oben

'Neue Ballposition und

'neue Bewegungsrichtung festlegen.

'Ball geht ueber Spielfeld hinaus:

'Zeile fuer Ball eingrenzen.

'Richtungsaenderung machen.

'Sound fuer Wand festlegen.

'Ball geht ueber Spielfeld hinaus:

'Spalte fuer Ball eingrenzen.

'Richtungsaenderung machen.

'Sound fuer Wand festlegen.

```

'Ziegel wurde getroffen:
If iZeileBall >= MAX_OBEN Then
    'Kontrolle ob die neue Position
    'Teil eines Ziegels ist:
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall, iSpalteBall) 'Wenn ja, dann Ziegel entfernen
        stRichtungNeu = "RU" 'und die Richtung aendern.
    End If
End If
    'Zusatzkontrolle falls Ziegel
    'von der Seite getroffen wird:
    If iZeileBall > MAX_OBEN Then
        If oTabellenblatt.Cells(iZeileBall - 1, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
            Call ZiegelLoeschen(iZeileBall - 1, iSpalteBall) 'Wenn ja, Ziegel entfernen
            stRichtungNeu = "RU" 'und die Richtung aendern.
        End If
    End If

Case "LO"
    iZeileBall = iZeileBall - 1
    iSpalteBall = iSpalteBall - 1
    'Ball bewegt sich nach: links oben
    'Neue Ballposition und
    'neue Bewegungsrichtung festlegen.
    If iZeileBall <= MAX_OBEN Then
        iZeileBall = MAX_OBEN
        stRichtungNeu = "LU"
        'Ball geht ueber Spielfeld hinaus:
        'Zeile fuer Ball eingrenzen.
        'Richtungsaenderung machen.
        stSoundAlias = ALIAS_SOUND_WAND 'Sound fuer Wand festlegen.
    End If
    If iSpalteBall <= MAX_LINKS Then
        iSpalteBall = MAX_LINKS
        stRichtungNeu = "RO"
        'Ball geht ueber Spielfeld hinaus:
        'Spalte fuer Ball eingrenzen.
        'Richtungsaenderung machen.
        stSoundAlias = ALIAS_SOUND_WAND 'Sound fuer Wand festlegen.
    End If

'Ziegel wurde getroffen:
If iZeileBall >= MAX_OBEN Then
    'Kontrolle ob die neue Position
    'Teil eines Ziegels ist:
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall, iSpalteBall) 'Wenn ja, dann Ziegel entfernen
        stRichtungNeu = "LU" 'und die Richtung aendern.
    End If
End If
    'Zusatzkontrolle falls Ziegel
    'von der Seite getroffen wird:
    If iZeileBall > MAX_OBEN Then
        If oTabellenblatt.Cells(iZeileBall - 1, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
            Call ZiegelLoeschen(iZeileBall - 1, iSpalteBall) 'Wenn ja, Ziegel entfernen
            stRichtungNeu = "LU" 'und die Richtung aendern.
        End If
    End If

Case "RU"
    iZeileBall = iZeileBall + 1
    iSpalteBall = iSpalteBall + 1
    'Ball bewegt sich nach: rechts unten
    'Neue Ballposition und
    'neue Bewegungsrichtung festlegen.
    If iZeileBall >= MAX_UNTEN Then
        iZeileBall = MAX_UNTEN
        stRichtungNeu = "RO"
        'Ball geht ueber Spielfeld hinaus:
        'Zeile fuer Ball eingrenzen.
        'Richtungsaenderung machen.
        bSoundWand = True 'Sound Wand ausgeben.
    End If
    If iSpalteBall >= MAX_RECHTS Then
        iSpalteBall = MAX_RECHTS
        stRichtungNeu = "LU"
        'Ball geht ueber Spielfeld hinaus:
        'Spalte fuer Ball eingrenzen.
        'Richtungsaenderung machen.
        bSoundWand = True 'Sound Wand ausgeben.
    End If

'Ball trifft auf Brett oder nicht:
If iZeileBall = MAX_UNTEN Then
    'Ball in Zeile von Brett:
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BRETT Then
        iZeileBall = iZeileBall - 1
        'Brett ist unter Ball:
        iSpalteBall = iSpalteBall - 1
        'Letzte Ballbewegung
    
```

```

stSoundAlias = ALIAS_SOUND_BRETT      'rueckgaengig machen und
Else                                   'Sound abspielen.
stSoundAlias = ALIAS_SOUND_ENDE        'Anderen Sound abspielen.
bBallImSpiel = False                  'Sonst ist Ball aus dem Spielfeld.
iBaelle = iBaelle - 1                  'Anzahl der Baelle verringern,
Call BlattschutzAufheben               'Blattschutz aufheben und
oTabellenblatt.Range("BAELLE") = iBaelle 'im Tabellenblatt ausgeben.
Call BlattschutzSetzen                 'Blattschutz wieder setzen.
If iBaelle > 0 Then                     'Sind noch Baelle im Spiel, dann
    Call SpielStarten(False)           'neuen Ball auf Brett legen.
Else                                    'Wenn kein Ball mehr im Spiel:
    stTastendruck = "SPIELEND"         'Abbruchsbedingung fuer
End If                                 'Endlos-Schleife setzen.
End If

Else                                   'Sound fuer Wand festlegen.
If bSoundWand = True Then stSoundAlias = ALIAS_SOUND_WAND
End If

'Ziegel wurde getroffen:
If iZeileBall <= MAX_UNTEN - 3 Then    'Kontrolle ob die neue Position
                                        'Teil eines Ziegels ist:
    If oTabellenblatt.Cells(iZeileBall + 1, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall + 1, iSpalteBall) 'Wenn ja, dann Ziegel entfernen
        stRichtungNeu = "RO"                               'und die Richtung aendern.
    End If
End If
If iZeileBall <= MAX_UNTEN - 3 Then    'Zusatzkontrolle falls Ziegel
                                        'von der Seite getroffen wird:
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall, iSpalteBall) 'Wenn ja, Ziegel entfernen
        stRichtungNeu = "RO"                               'und die Richtung aendern.
    End If
End If

Case "LU"                             'Ball bewegt sich nach: links unten
iZeileBall = iZeileBall + 1            'Neue Ballposition und
iSpalteBall = iSpalteBall - 1          'neue Bewegungsrichtung festlegen.
If iZeileBall >= MAX_UNTEN Then         'Ball geht ueber Spielfeld hinaus:
    iZeileBall = MAX_UNTEN              'Zeile fuer Ball eingrenzen.
    stRichtungNeu = "LO"               'Richtungsänderung machen.
    bSoundWand = True                  'Sound Wand ausgeben.
End If
If iSpalteBall <= MAX_LINKS Then        'Ball geht ueber Spielfeld hinaus:
    iSpalteBall = MAX_LINKS            'Spalte fuer Ball eingrenzen.
    stRichtungNeu = "RU"               'Richtungsänderung machen.
    bSoundWand = True                  'Sound Wand ausgeben.
End If

'Ball trifft auf Brett oder nicht:
If iZeileBall = MAX_UNTEN Then          'Ball in Zeile von Brett:
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BRETT Then
        iZeileBall = iZeileBall - 1    'Brett ist unter Ball:
        iSpalteBall = iSpalteBall + 1  'Letzte Ballbewegung
        stSoundAlias = ALIAS_SOUND_BRETT 'rueckgaengig machen und
    Else                                'Sound abspielen.
        stSoundAlias = ALIAS_SOUND_ENDE 'rueckgaengig machen und
        bBallImSpiel = False           'Sonst ist Ball aus dem Spielfeld.
        iBaelle = iBaelle - 1          'Anzahl der Baelle verringern,
        Call BlattschutzAufheben       'Blattschutz aufheben und
        oTabellenblatt.Range("BAELLE") = iBaelle 'im Tabellenblatt ausgeben.
        Call BlattschutzSetzen         'Blattschutz wieder setzen.
        If iBaelle > 0 Then             'Sind noch Baelle im Spiel, dann
            Call SpielStarten(False)   'neuen Ball auf Brett legen.
        Else                            'Wenn kein Ball mehr im Spiel:
            stTastendruck = "SPIELEND" 'Abbruchsbedingung fuer
        End If
    End If
End If

```

```

End If                                     'Endlos-Schleife setzen.
End If
Else                                       'Sound fuer Wand festlegen.
    If bSoundWand = True Then stSoundAlias = ALIAS_SOUND_WAND
End If

'Ziegel wurde getroffen:
If iZeileBall <= MAX_UNTEN - 3 Then      'Kontrolle ob die neue Position
                                         'Teil eines Ziegels ist:
    If oTabellenblatt.Cells(iZeileBall + 1, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall + 1, iSpalteBall) 'Wenn ja, dann Ziegel entfernen
        stRichtungNeu = "LO"                               'und die Richtung aendern.
    End If
End If                                     'Zusatzkontrolle falls Ziegel
                                         'von der Seite getroffen wird:
If iZeileBall <= MAX_UNTEN - 3 Then
    If oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color <> FARBE_HINTERGRUND
Then
        Call ZiegelLoeschen(iZeileBall, iSpalteBall) 'Wenn ja, Ziegel entfernen
        stRichtungNeu = "LO"                               'und die Richtung aendern.
    End If
End If
End Select                               'Zum Schluss noch den Ball auf
                                         'die neue Position setzen und
oTabellenblatt.Cells(iZeileBall, iSpalteBall).Interior.Color = FARBE_BALL
stRichtung = stRichtungNeu              'die neue Richtung festlegen.
Range("CURSOR_PARKPOSITION").Select     'Cursor auf Parkposition setzen.

If stSoundAlias <> "" Then Call SoundAbspielen(stSoundAlias) 'Sound wiedergeben.

Beenden:                                'Fehlerbehandlungsroutine:
                                         '=====
Application.Cursor = xlDefault          'Sanduhr ausblenden.
Exit Function                           'Funktion abbrechen.

Fehler:                                 'Fehlermarke - Fehlerbehandlung:
NachLinks = False                      'Returnwert = Fehler setzen.
MsgBox Err.Description                 'Fehlermeldung ausgeben.
Resume Beenden                         'Bei Funktionsende weitermachen.
Resume                                 'Nur fuer Testzwecke hinterlegt.
End Function                           'Funktionsende.
    
```

4.4.5.) Das Spiel mit Sounds hinterlegen

Die Hinterlegung von Sounds haben wir in der Übungseinheit #2 bereits im Detail abgehandelt. Daher wird hier nur mehr das Wichtigste, das Spiel Bricks betreffend, abgehandelt.

Zuerst müssen wir uns überlegen bei welchen Aktionen ein Sound wiedergegeben werden soll und vor allem welcher Sound. Bei Bricks würde ich folgendes vorschlagen:

Hintergrundmusik: Hintergrundmusik.wav.

Wenn Ball auf das Brett kommt – ein verzerrtes Piepsen – AufBrett.mp3
 Wenn Ball auf einen Ziegel trifft – ein dumpfes Piepsen – AufBrick.mp3
 Wenn Ball an die Wand schlägt – ein helles Piepsen – AufWand.mp3
 Wenn Ball das Brett verfehlt – ein tiefer Piepstön – AusSpiel.mp3

Die Piepstöne habe ich aus dem Internet, von einer Seite welche hunderte Piepstöne zur Verfügung stellt, heruntergeladen. Gefunden habe ich die Seite bei der Suche nach: Piepstöne.

Nun ergänzen wir uns im Modul **Globales** die Variablen und Konstanten für die Sounds:

```
'-----*/
'- Globale Konstanten und Variablen - fuer Hintergrundmusik und Sounds -*/
'-----*/

Global bSoundsAbspielen           As Boolean   'Sounds abspielen lassen.
Global bSoundEndeOK               As Boolean   'Sound AusSpiel kann abgespielt werden.
Global bSoundBrickOK              As Boolean   'Sound AufBrick kann abgespielt werden.
Global bSoundBrettOK              As Boolean   'Sound AufBrett kann abgespielt werden.
Global bSoundWandOK               As Boolean   'Sound AufWand kann abgespielt werden.
Global stDateiHintergrundMusik    As String    'Akt. Dateiname fuer Hintergrundmusik.

Global Const DATEI_SOUND_ENDE = "\AusSpiel.mp3"
Global Const DATEI_SOUND_BRICK = "\AufBrick.mp3"
Global Const DATEI_SOUND_BRETT = "\AufBrett.mp3"
Global Const DATEI_SOUND_WAND = "\AufWand.mp3"
Global Const DATEI_SOUND_HINTERGRUND = "\Hintergrundmusik.wav"

Global Const ALIAS_SOUND_ENDE = "Alias_Sound_End"
Global Const ALIAS_SOUND_BRICK = "Alias_Sound_Brick"
Global Const ALIAS_SOUND_BRETT = "Alias_Sound_Brett"
Global Const ALIAS_SOUND_WAND = "Alias_Sound_Wand"
Global Const ALIAS_SOUND_HINTERGRUND = "Alias_Hintergrundmusik"
```

Dann behandeln wir die Schaltfläche **Ton EIN**. Dazu ergänzen wir die Prozedur **Workbook_Open** in **DieseArbeitsmappe** mit folgendem Programmcode:

```
'*-----*/
Set oTabellenblatt = Sheets(TABELLENBLATT)           'Akt. Tabellenblatt festlegen.
On Error Resume Next                                 'Bei Fehler einfach weitermachen,
                                                    'da nicht alle Shapes die
                                                    'Eigenschaft TextFrame haben.
For i = 0 To oTabellenblatt.Shapes.Count              'Alle Shapes (Frames) durchsuchen:
                                                    'Sollte angezeigt werden, den
    If oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton AUS" Then 'Ton NICHT
        oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton EIN" 'wiedergeben,
    End If                                           'dann Text umaendern, damit der
Next i                                              'Spieler Ton selbst aktivieren kann.
On Error GoTo Fehler                                'Bei Fehler wieder zur Fehlermarke.
bSoundsAbspielen = False                            'Musikwiedergabe deaktivieren.

lPunkte = 0                                           'Punkteanzahl 0 setzen.
Call SpielStarten(True)                              'Startprogramm aufrufen.
'*-----*/
```

Anschließend schreiben wir im Modul **Makros** das Makro **Ton_Umschalten** welches wir der Schaltfläche zuweisen (sollte eigentlich bereits in der Mustervorlage hinterlegt sein, wenn nicht – dann später ergänzen):

```
Public Sub Ton_Umschalten()
'*****/
'* Makro wird ausgefuehrt bei Klick auf Schaltflaeche: Ton AUS bzw. Ton EIN*/
'*****/
Call TonUmschalten
End Sub                                     'Prozedurende.
```

Genauso verfahren wir mit der Funktion **TonUmschalten** im Modul **Sounds**:

```
Public Function TonUmschalten()
'***** INTERNES UNTERPROGRAMM - INTERNE FUNKTION *****/
'*          INTERNES UNTERPROGRAMM - INTERNE FUNKTION          */
'*****
'*-----*/
'*          */
'*          FUNKTION:      TonUmschalten          */
'*          */
'*          BESCHREIBUNG:   Schaltet die Wiedergabe von Hintergrundmusik und */
'*          Geraeuschen aus bzw. ein.          */
'*          */
'*          RETURN         nichts          */
'*          */
'*          PARAMETER:     keine          */
'*          */
'*-----*/
Dim i                As Integer                'Allgemeine Zaehlvariable.

Set oTabellenblatt = Sheets(TABELLENBLATT)    'Akt. Tabellenblatt festlegen.
oTabellenblatt.Select                        'Tabellenblatt auswaehlen.

'Sollte ein Shape (Form) nicht die Eigenschaft: TextFrame.Characters.Text
'haben, dann tritt ein Laufzeitfehler auf. Um dieses zu verhindern wird
'der Programmcode bei einem Fehler in der naechsten Programmzeile fortgesetzt.
'Genaugenommen wird das naechste Shape abgehandelt.

On Error Resume Next
For i = 0 To oTabellenblatt.Shapes.Count
    Err = 0
    If oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton AUS" Then
        If Err = 0 Then
            bSoundsAbspielen = False
            Call SpielSoundsRuecksetzen
            Call HintergrundMusik_AUS
            oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton EIN"
            Exit For
        End If
    ElseIf oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton EIN" Then
        If Err = 0 Then
            bSoundsAbspielen = True
            Call SpielSoundsDefinieren
            Call HintergrundMusik_EIN
            oTabellenblatt.Shapes(i).TextFrame.Characters.Text = "Ton AUS"
            Exit For
        End If
    Else
        'nix machen
    End If
Next i
End Function
```

'Bei Fehler einfach weitermachen:
'Alle Formen im Tabellenblatt
'durcharbeiten. Fehler 0 setzen.
'Wenn Ton ausschalten:
'Merken: Keine Tonwiedergabe.
'Alle Sounds zuruecksetzen.
'Hintergrundmusik stoppen.
'Beschriftung aendern und
'Funktion beenden.
'Wenn Ton einschalten:
'Merken: Ton wiedergeben.
'Alle Sounds neu definieren.
'Hintergrundmusik starten.
'Beschriftung aendern und
'Funktion beenden.
'Reserve fuer Erweiterungen:
'Funktionsende.

Als nächstes überarbeiten wir die Funktionen **SpielSoundsDefinieren**, **HintergrundMusik_EIN**, **HintergrundMusik_AUS**, **SoundAbspielen** und **SpielSoundsRuecksetzen** im Modul **Sounds**:

```
Public Function SpielSoundsDefinieren() As Boolean
'***** INTERNES UNTERPROGRAMM - INTERNE FUNKTION *****/
'*          INTERNES UNTERPROGRAMM - INTERNE FUNKTION          */
'*****
'*-----*/
'*          */
'*          FUNKTION:      SpielSoundsDefinieren          */
'*          */
'*          BESCHREIBUNG:   Definiert die Aliasnamen von MP3-Sounddateien.  */
'*          */
'*-----*/
```



```

'*
'*      RETURN          Boolean ..... True = Funktion in Ordnung
'*                                     False = Fehler in Funktion
'*      PARAMETER:      keine
'*
'*-----*/
Dim stDateiname          As String          'Sound-Datei.
Dim stAlias              As String          'Aliasname fuer Sound-Datei.
Dim stBuffer             As String          'Buffer fuer DOS 8.3 Dateinamen.

On Error GoTo Fehler          'Bei Fehler zur Fehlerbehandlung.
SpielSoundsDefinieren = True  'Returnwert = alles OK setzen.
Call SpielSoundsRuecksetzen   'Zuerst alles ruecksetzen.

'*-----*/
'*- Sound nachdem der Ball aus dem Spielfeld ist
'*-----*/
stDateiname = ActiveWorkbook.Path & DATEI_SOUND_ENDE 'Dateiname für Sound und
stAlias = ALIAS_SOUND_ENDE                          'Namen fuer Alias festlegen.
stBuffer = Space$(255)                               'DOS 8.3 Dateinamen ermitteln:
If GetShortPathName(stDateiname, stBuffer, Len(stBuffer)) <> 0 Then
    stDateiname = Left$(stBuffer, InStr(stBuffer, vbNullChar) - 1)
End If
If PlayMP3Sound("open " & stDateiname & " type MPEGVideo alias " & stAlias, 0, 0, 0) = 0 Then
    bSoundEndeOK = True
Else
    bSoundEndeOK = False
End If

'*-----*/
'*- Sound nachdem der Ball auf einen Ziegel stoest
'*-----*/
stDateiname = ActiveWorkbook.Path & DATEI_SOUND_BRICK 'Dateiname für Sound und
stAlias = ALIAS_SOUND_BRICK                          'Namen fuer Alias festlegen.
stBuffer = Space$(255)                               'DOS 8.3 Dateinamen ermitteln:
If GetShortPathName(stDateiname, stBuffer, Len(stBuffer)) <> 0 Then
    stDateiname = Left$(stBuffer, InStr(stBuffer, vbNullChar) - 1)
End If
If PlayMP3Sound("open " & stDateiname & " type MPEGVideo alias " & stAlias, 0, 0, 0) = 0 Then
    bSoundBrickOK = True
Else
    bSoundBrickOK = False
End If

'*-----*/
'*- Sound nachdem der Ball auf das Brett kommt
'*-----*/
stDateiname = ActiveWorkbook.Path & DATEI_SOUND_BRETT 'Dateiname für Sound und
stAlias = ALIAS_SOUND_BRETT                          'Namen fuer Alias festlegen.
stBuffer = Space$(255)                               'DOS 8.3 Dateinamen ermitteln:
If GetShortPathName(stDateiname, stBuffer, Len(stBuffer)) <> 0 Then
    stDateiname = Left$(stBuffer, InStr(stBuffer, vbNullChar) - 1)
End If
If PlayMP3Sound("open " & stDateiname & " type MPEGVideo alias " & stAlias, 0, 0, 0) = 0 Then
    bSoundBrettOK = True
Else
    bSoundBrettOK = False
End If

'*-----*/
'*- Sound nachdem der Ball an die Wand trifft
'*-----*/
stDateiname = ActiveWorkbook.Path & DATEI_SOUND_WAND 'Dateiname für Sound und
stAlias = ALIAS_SOUND_WAND                          'Namen fuer Alias festlegen.
stBuffer = Space$(255)                               'DOS 8.3 Dateinamen ermitteln:
If GetShortPathName(stDateiname, stBuffer, Len(stBuffer)) <> 0 Then
    stDateiname = Left$(stBuffer, InStr(stBuffer, vbNullChar) - 1)

```

```

End If
If PlayMP3Sound("open " & stDateiname & " type MPEGVideo alias " & stAlias, 0, 0, 0) = 0 Then
    bSoundWandOK = True
Else
    bSoundWandOK = False
End If

'-----*/
'*- Hintergrundmusik -*/
'-----*/
stDateiHintergrundMusik = ActiveWorkbook.Path & DATEI_SOUND_HINTERGRUND 'Dateinamen für
                                'Hintergrundmusik 2 festlegen.
stBuffer = Space$(255) 'DOS 8.3 Dateinamen ermitteln:
If GetShortPathName(stDateiHintergrundMusik, stBuffer, Len(stBuffer)) <> 0 Then
    stDateiHintergrundMusik = Left$(stBuffer, InStr(stBuffer, vbNullChar) - 1)
Else
    stDateiHintergrundMusik = ""
End If

'Fehlerbehandlungsroutine:
'=====
Beenden:
    Exit Function
'Funktion abbrechen.

Fehler:
    SpielSoundsDefinieren = False
    MsgBox Err.Description
    Resume Beenden
    Resume
End Function
'Fehlermarke - Fehlerbehandlung:
'Returnwert = Fehler setzen.
'Fehlermeldung ausgeben.
'Bei Funktionsende weitermachen.
'Nur fuer Testzwecke hinterlegt.
'Funktionsende.

Public Function HintergrundMusik_EIN()
'*****
'*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      */
'*****
'-----*/
'*      */
'*      FUNKTION:      HintergrundMusik_EIN      */
'*      */
'*      BESCHREIBUNG:   Startet Wiedergabe der Hintergrundmusik in einer */
'*      Endlos-Schleife.      */
'*      */
'*      RETURN          nichts      */
'*      */
'*      PARAMETER:      keine      */
'*      */
'-----*/
If bSoundsAbspielen = True And stDateiHintergrundMusik <> "" Then 'Wenn abgespielt
    Call PlayWaveSound(stDateiHintergrundMusik, SND_ASYNC Or SND_LOOP) 'werden soll und
End If
End Function
'Datei fuer Wiedergabe vorhanden.
'Funktionsende.

Public Function HintergrundMusik_AUS()
'*****
'*      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      */
'*****
'-----*/
'*      */
'*      FUNKTION:      HintergrundMusik_AUS      */
'*      */
'*      BESCHREIBUNG:   Stoppt Wiedergabe der Hintergrundmusik.      */
'*      */
'*      RETURN          nichts      */
'*      */
'*      PARAMETER:      keine      */
'*      */
'-----*/
Call PlayWaveSound(0, SND_PURGE)
End Function
'Wiedergabe sofort beenden.
'Funktionsende.
    
```



```

Public Function SoundAbspielen(stAliasname As String)
'***** SoundAbspielen(stAliasname As String) *****/
' *      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      */
'*****/
' *-----*/
' *      */
' *      FUNKTION:      SoundAbspielen      */
' *      */
' *      BESCHREIBUNG:  Startet Wiedergabe der entsprechenden Datei zum */
' *                      uebergebenen Alias-Namen.      */
' *      */
' *      RETURN      nichts      */
' *      */
' *      PARAMETER:    String ..... Aliasname fuer abzuspielender */
' *                      Geraeusch.      */
' *      */
' *-----*/

If bSoundsAbspielen = True Then      'Wenn abgespielt werden soll:
Select Case stAliasname      'Je nachdem welches Geraeusch:
Case ALIAS_SOUND_ENDE      'Wenn dazugehoerige Datei
If bSoundEndeOK = True Then      'abgespielt werden kann, dann
Call PlayMP3Sound("play " & stAliasname & " from 0", 0, 0, 0)
End If      'diese abspielen lassen.

Case ALIAS_SOUND_BRICK      'Wenn dazugehoerige Datei
If bSoundBrickOK = True Then      'abgespielt werden kann, dann
Call PlayMP3Sound("play " & stAliasname & " from 0", 0, 0, 0)
End If      'diese abspielen lassen.

Case ALIAS_SOUND_BRETT      'Wenn dazugehoerige Datei
If bSoundBrettOK = True Then      'abgespielt werden kann, dann
Call PlayMP3Sound("play " & stAliasname & " from 0", 0, 0, 0)
End If      'diese abspielen lassen.

Case ALIAS_SOUND_WAND      'Wenn dazugehoerige Datei
If bSoundWandOK = True Then      'abgespielt werden kann, dann
Call PlayMP3Sound("play " & stAliasname & " from 0", 0, 0, 0)
End If      'diese abspielen lassen.

Case Else

End Select

End If

End Function      'Funktionsende.

```

```
Public Function SpielSoundsRuecksetzen()  
'*****  
' *      INTERNES UNTERPROGRAMM - INTERNE FUNKTION      *  
'*****  
' *-----*  
' *  
' *      FUNKTION:      SpielSoundsRuecksetzen      *  
' *  
' *      BESCHREIBUNG:   Stoppt alle Wiedergaben und schließt MCI's.      *  
' *  
' *      RETURN          nichts      *  
' *  
' *      PARAMETER:      keine      *  
' *-----*  
PlayMP3Sound "stop" & ALIAS_SOUND_ENDE, 0, 0, 0      'Wiedergabe stoppen.  
PlayMP3Sound "close" & ALIAS_SOUND_ENDE, 0, 0, 0      'MCI schliessen.  
  
PlayMP3Sound "stop" & ALIAS_SOUND_BRICK, 0, 0, 0      'Wiedergabe stoppen.  
PlayMP3Sound "close" & ALIAS_SOUND_BRICK, 0, 0, 0      'MCI schliessen.
```

Klickomania

```
PlayMP3Sound "stop " & ALIAS_SOUND_BRETT, 0, 0, 0 'Wiedergabe stoppen.  
PlayMP3Sound "close " & ALIAS_SOUND_BRETT, 0, 0, 0 'MCI schliessen.  
  
PlayMP3Sound "stop " & ALIAS_SOUND_WAND, 0, 0, 0 'Wiedergabe stoppen.  
PlayMP3Sound "close " & ALIAS_SOUND_WAND, 0, 0, 0 'MCI schliessen.  
  
End Function 'Funktionsende.
```

Zum Schluß binden wir die Piepstöne noch ins Spiel ein. Dazu ergänzen wir in der Funktion **ZiegelLoeschen**, gleich als erste Programmzeile, mit folgendem Code:

```
Call SoundAbspielen(ALIAS_SOUND_BRICK) 'Sound abspielen.
```

Die weiteren Piepstöne werden in der Funktion BallBewegen abgespielt. Da 3 unterschiedliche Piepser abgespielt werden können, stellen wir zuerst fest, ob und welcher Piepston ausgegeben werden soll. Dazu speichern wir den Aliasnamen des Sounds in einer Variablen. Beinhaltet am Ende der Funktion die Variable einen Wert, dann wird dieser Sound abgespielt.

Die benötigten Ergänzungen wurden bereits im vorherigen Kapitel abgebildet und die betreffenden Programmzeilen gelb hervorgehoben.

Kontrollieren bzw. ergänzen Sie in der Prozedur **Workbook_BeforeClose** und **Workbook_Deactivate** ob die Hintergrundmusik auch wieder ausgeschaltet wird:

```
Call HintergrundMusik_AUS 'Hintergrundmusik beenden.
```

Kontrollieren bzw. ergänzen Sie in der Prozedur **Workbook_Activate** ob die Hintergrundmusik wenn nötig auch wieder eingeschaltet wird:

```
If bSoundsAbspielen = True Then 'Wenn Soundwiedergabe aktiviert:  
    Call HintergrundMusik_EIN 'Hintergrundmusik wiedergeben.  
End If
```


5.) Schlußbemerkungen

Für ein Spiel sollten Sie auch ein **Icon** erstellen, welches man einer Verknüpfung zuweisen kann. Zum Erstellen eigener Icons verwende ich den kostenlosen **Greenfish Icon Editor Pro**. Damit wurde auch das Icon für das Bricks -Spiel erstellt:



Bei komplexeren Spielen sollte auch eine detaillierte **Spielbeschreibung** erstellt werden.

Definieren Sie den Schutz für die Zellen im Tabellenblatt. Am besten zuerst alle Zellen markieren und dann den Schutz für alle Zellen aktivieren:




Danach jene Zellen auswählen, welche vom Programm beschrieben werden, hier das gesamte Spielfeld (Zellen müssen angeklickt werden können), den Titel (CURSOR_PARKPOSITION) und die Zelle mit der Punkteanzahl, und dort das Häkchen bei "Gesperrt" entfernen.

Aktivieren Sie den Blattschutz in Excel. Im Register **Überprüfen**, Symbol **Blatt schützen**:

Beim Klickomania müssen die Berechtigung: **Nicht gesperrte Zellen auswählen** (da wir die Zellen ja anklicken wollen), **Zellen formatieren** (da wir die Hintergrundfarbe der Steine verändern) und **Objekte bearbeiten** (da wir die Beschriftung in einem Shape ändern) abgehakt werden.

Schützen Sie zusätzlich Ihr **VBA-Projekt** mit einem Paßwort wenn Sie den Programmcode geheim halten wollen. In Visual Basic das VBA-Projekt mit der rechten Maustaste anklicken -> Eigenschaften von VBA-Projekt... -> Registerblatt: Schutz. Dort zweimal das Kennwort eingeben und Klick auf OK.



So das wär's dann wieder Mal – viel Spaß beim Nachprogrammieren oder Überarbeiten des Spieles.